

Power analysis and sample-size determination in survival models with the new **stpower** command

Yulia Marchenko

Senior Statistician
StataCorp LP

2007 Boston Stata Users Group Meeting



- 1 Introduction
 - Theory and terminology
 - Introduction to **stpower** subcommands
- 2 Sample-size determination for survival studies
 - Log-rank test
 - Cox proportional hazards model
 - Exponential survivor functions
- 3 Power and effect-size determination
- 4 Tabulating results
 - Default tables
 - Customized tables
- 5 Example of using a dialog box
- 6 Power and other curves
 - Manual generation of power and other curves
 - Automatic generation of power and other curves
- 7 Conclusion

- type I study
all subjects experience an event (fail) by the end of the study
- type II study
a study terminates after a fixed time T resulting in censored subjects

- type I study
all subjects experience an event (fail) by the end of the study
- type II study
a study terminates after a fixed time T resulting in censored subjects
- administrative censoring
the right censoring occurring when the study observation period ends
- loss to follow-up (withdrawal), $L(t)$
occurs when subjects fail to complete the course of the study for reasons unrelated to the event of interest

- accrual period, R
period during which subjects are being enrolled into a study
- follow-up period, f
period during which subjects are under observation and no new subjects enter a study

- accrual period, R
period during which subjects are being enrolled into a study
- follow-up period, f
period during which subjects are under observation and no new subjects enter a study
- exponential (parametric) test
test on difference or log ratio of (two) exponential hazard rates

Main components of sample-size computation

Choose:

- method of analysis
log-rank test, Cox PH model, parametric test
- probability of a type I error (significance level), α
usually 0.05, 0.01
- power, $1 - \beta$ (or probability of a type II error, β)
usually 80%, 90% (or 20%, 10%)
- effect size, ψ , usually expressed as
the hazard ratio, $\Delta = h_2(t)/h_1(t), \forall t$ (PH assumption), or
the log of the hazard ratio, $\ln \Delta$, or
the coefficient in a Cox regression model,
$$\beta_1 = \ln\{h(t, x_1 + 1|x_2, \dots, x_p)/h(t, x_1|x_2, \dots, x_p)\}$$

General formula

The general formula for the required number of subjects in a survival study may be expressed as

$$N = \frac{E(\alpha, \beta, \psi)}{p_E\{S(t), L(t), R, T\}}$$

where

$E()$ is the number of events required to be observed in a study, and $p_E()$ is the probability of observing an event in a study.

Note:

- power, $1 - \beta$, is directly related to the number of events
- duration of a study T , accrual and loss to follow up patterns affect the probability of a subject to experience an event in a study.

Three subcommands:

- `stpower logrank`
- `stpower cox`
- `stpower exponential`

All subcommands:

- share main common options `alpha()`, `power()`, `beta()`, `n()`, `hratio()`, `onesided`, ...
- accept multiple values (*numlist*) in main options
- facilitate customizable tables of results
- save results in a dataset

- Type of test
 - two-sample log-rank test
- Computation
 - sample size (given power and hazard ratio)
 - power (given sample size and hazard ratio)
 - (log) hazard ratio (given power and sample size)
- Capabilities
 - unequal group allocation
 - uniform accrual
 - conservative adjustment for withdrawal
- Methodology
 - Freedman (1982) (default)
 - Schoenfeld (1981) (option schoenfeld)

Sample-size determination for the log-rank test

Objective. Obtain the required sample size to ensure prespecified power of a two-sided log-rank test at level α to detect a $\Delta_a \times 100\%$ reduction in hazard of the experimental group relative to the control group.

Hypothesis. $H_o: S_1(t) = S_2(t)$ vs $H_a: S_1(t) \neq S_2(t)$

Assumptions. Proportional hazards, $S_2(t) = \{S_1(t)\}^\Delta$; large-sample approximation to the test statistic holds

Equivalent hypothesis. $H_o: \Delta = 1$ vs $H_a: \Delta \neq 1$ (default)
 $H_o: \ln(\Delta) = 0$ vs $H_a: \ln(\Delta) \neq 0$ (schoenfeld)

Example

- Estimate required sample size to achieve 80% power to detect 50% reduction in a hazard of the experimental group by using a two-sided 0.05-level log-rank test under 4 different study designs (A, B, C, and D below).
- All of these study designs assume $\alpha = 0.05$, $1 - \beta = 0.8$, and $\Delta_a = 0.5$.

Design A: type I study (unlimited follow up); 1:1 randomization.

```
. stpower logrank, power(0.8) hratio(0.5) nratio(1)
```

Estimated sample sizes for two-sample comparison of survivor functions

Log-rank test, Freedman method

Ho: $S_1(t) = S_2(t)$

Input parameters:

```
alpha =    0.0500  (two sided)
hratio =    0.5000
power =    0.8000
p1 =       0.5000
```

Estimated number of events and sample sizes:

```
E =        72
N =        72
N1 =       36
N2 =       36
```

Design B: type II study (40% of subjects in the control group survive by the end of the study); 1:1 randomization; no withdrawal.

```
. stpower logrank 0.4
```

```
Estimated sample sizes for two-sample comparison of survivor functions  
Log-rank test, Freedman method
```

```
Ho:  $S_1(t) = S_2(t)$ 
```

```
Input parameters:
```

```
alpha = 0.0500 (two sided)  
s1 = 0.4000  
s2 = 0.6325  
hratio = 0.5000  
power = 0.8000  
p1 = 0.5000
```

```
Estimated number of events and sample sizes:
```

```
E = 72  
N = 148  
N1 = 74  
N2 = 74
```

Design C: type II study (40% of subjects in the control group survive by the end of the study); 1:2 randomization; no withdrawal.

```
. stpower logrank 0.4, nratio(2)
```

```
Estimated sample sizes for two-sample comparison of survivor functions  
Log-rank test, Freedman method
```

```
Ho: S1(t) = S2(t)
```

```
Input parameters:
```

```
alpha = 0.0500 (two sided)  
s1 = 0.4000  
s2 = 0.6325  
hratio = 0.5000  
power = 0.8000  
p1 = 0.3333
```

```
Estimated number of events and sample sizes:
```

```
E = 63  
N = 142  
N1 = 47  
N2 = 95
```

Design D: type II study (40% of subjects in the control group survive by the end of the study); 1:2 randomization; 10% withdrawal.

```
. stpower logrank 0.4, nratio(2) wdprob(0.1)
Estimated sample sizes for two-sample comparison of survivor functions
Log-rank test, Freedman method
Ho: S1(t) = S2(t)
Input parameters:
      alpha =      0.0500  (two sided)
        s1 =      0.4000
        s2 =      0.6325
    hratio =      0.5000
      power =      0.8000
        p1 =      0.3333
withdrawal =     10.00%
Estimated number of events and sample sizes:
      E =         63
      N =        157
     N1 =         52
     N2 =        105
```


- Type of test
 - Wald test of a covariate in a Cox PH model
- Computation
 - sample size (given power and coefficient)
 - power (given sample size and coefficient)
 - coefficient (hazard ratio) (given power and sample size)
- Capabilities
 - binary or continuous covariate
 - adjustment for other covariates in a model
 - conservative adjustment for withdrawal
- Methodology
 - Schoenfeld (1983), Hsieh and Lavori (2000)

Objective. Obtain the required sample size to ensure prespecified power of a two-sided α -level Wald test to detect a change of $\beta_{1a} = \ln(\Delta_a)$ in log hazards for a one-unit change in a covariate of interest x_1 adjusted for other factors x_2, \dots, x_p .

Hypothesis. $H_0: (\beta_1, \beta_2, \dots, \beta_p) = (0, \beta_2, \dots, \beta_p)$ vs
 $H_a: (\beta_1, \beta_2, \dots, \beta_p) = (\beta_{1a}, \beta_2, \dots, \beta_p)$

Assumptions. Proportional hazards; large-sample approximation to the test statistic holds.

Example

- Hsieh and Lavori (2000) estimate required sample size for a study of multiple-myeloma patients investigating the effect of the log of the blood urea nitrogen, IBUN, on patients' survival.
- The significance of the effect is to be determined via a one-sided Wald test on the coefficient of IBUN, β_1 , estimated from the Cox model in the presence of other covariates.
- Main study parameters: $\alpha = 0.05$, $1 - \beta = 0.8$, $\beta_{1a} = 1$, $\sigma = 0.3126$.
- We'll consider several study designs.

Assumptions: IBUN is independent of other covariates ($R^2 = 0$);
no censoring ($p_E = 1$).

```
. stpower cox 1, sd(0.3126) onesided  
  
Estimated sample size for Cox PH regression  
Wald test, log-hazard metric  
Ho: [b1, b2, ..., bp] = [0, b2, ..., bp]  
  
Input parameters:  
      alpha =    0.0500  (one sided)  
      b1 =    1.0000  
      sd =    0.3126  
      power =    0.8000  
  
Estimated number of events and sample size:  
      E =    64  
      N =    64
```

Assumptions: IBUN is not independent of other covariates ($R^2 = 0.1837$);
no censoring ($p_E = 1$).

```
. stpower cox 1, sd(0.3126) onesided r2(0.1837)
```

```
Estimated sample size for Cox PH regression
```

```
Wald test, log-hazard metric
```

```
Ho: [b1, b2, ..., bp] = [0, b2, ..., bp]
```

```
Input parameters:
```

```
alpha = 0.0500 (one sided)
```

```
b1 = 1.0000
```

```
sd = 0.3126
```

```
power = 0.8000
```

```
R2 = 0.1837
```

```
Estimated number of events and sample size:
```

```
E = 78
```

```
N = 78
```

Assumptions: IBUN is not independent of other covariates ($R^2 = 0.1837$);
the overall death rate is $p_E = 0.738$.

```
. stpower cox 1, sd(0.3126) onesided r2(0.1837) failprob(0.738)
```

Estimated sample size for Cox PH regression

Wald test, log-hazard metric

Ho: [b1, b2, ..., bp] = [0, b2, ..., bp]

Input parameters:

```
alpha = 0.0500 (one sided)
b1 = 1.0000
sd = 0.3126
power = 0.8000
Pr(event) = 0.7380
R2 = 0.1837
```

Estimated number of events and sample size:

```
E = 78
N = 106
```

- Type of test
 - two-sample exponential test on difference of hazards or log hazards (option `loghazard`)
- Computation
 - sample size (given power and difference (or ratio) of hazards)
 - power (given sample size and difference (or ratio) of hazards)
- Capabilities
 - unequal group allocation
 - uniform or truncated exponential accrual
 - group-specific exponential loss to follow up hazard rates
 - conditional or unconditional approaches
- Methodology
 - Lachin (1981), Lachin and Foulkes (1986) (default)
 - Rubinstein, Gail and Santner (1981) (options `loghazard` and `unconditional`)

Objective. Obtain the required sample size to ensure prespecified power of a test of disparity in two exponential survivor functions with hazard rates λ_1 and λ_2 .

The disparity may be expressed as a difference between the hazard or log hazard rates, $\delta = \lambda_2 - \lambda_1$ or $\ln(\Delta) = \ln(\lambda_2) - \ln(\lambda_1)$, resp.

Assumptions. Exponential survivor functions, $S_i(t) = \exp\{-\lambda_i t\}$, $i = 1, 2$; large-sample approximation to the test statistic holds

Equivalent hypothesis. $H_o: \delta = 0$ vs $H_a: \delta \neq 0$ (default)
 $H_o: \ln(\Delta) = 0$ vs $H_a: \ln(\Delta) \neq 0$ (loghazard)

Example

- Lachin (2000, 412) demonstrates sample-size determination for a study comparing two therapies for lupus nephritis.
- From previous studies control-group survivor function was found to be log-linear with constant yearly hazard rate $\lambda_1 = 0.3$ ($t_{50} \approx 2.31$).
- Study parameters: $\alpha = 0.05$ (one sided), $1 - \beta = 0.9$,
 $\delta = \lambda_2 - \lambda_1 = -0.15$

Design A: no right censoring (unlimited follow up).

```
. stpower exponential 0.3 0.15, onesided power(0.9)
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Input parameters:

```
alpha =    0.0500  (one sided)
h1 =      0.3000
h2 =      0.1500
h2-h1 =   -0.1500
power =    0.9000
p1 =      0.5000
```

Estimated sample sizes:

```
N =        82
N1 =       41
N2 =       41
```

Design B: fixed-duration study ($R = 4, f = 2$); uniform accrual.

```
. stpower exponential 0.3 0.15, onesided power(0.9) aperiod(4) fperiod(2)
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Input parameters:

alpha = 0.0500 (one sided)

h1 = 0.3000

h2 = 0.1500

h2-h1 = -0.1500

power = 0.9000

p1 = 0.5000

Accrual and follow-up information:

duration = 6.0000

follow-up = 2.0000

accrual = 4.0000 (uniform)


Estimated sample sizes:

N = 136

N1 = 68

N2 = 68

Design C: fixed-duration study ($R = 4$, $f = 2$); truncated exponential accrual with shape -2 .



```
. stpower exponential 0.3 0.15, onesided power(0.9) aperiod(4) fperiod(2)
> ashape(-2)
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions
Exponential test, hazard difference, conditional
Ho: $h_2 - h_1 = 0$

Input parameters:

(omitted)

Accrual and follow-up information:

```
duration =      6.0000
follow-up =      2.0000
accrual =      4.0000 (exponential)
accrued(%) =     50.00 (by time t*)
t* =      3.6536 (91.34% of accrual)
```

Estimated sample sizes:

```
N =      184
N1 =      92
N2 =      92
```

Design D: fixed-duration study ($R = 4$, $f = 2$); truncated exponential accrual with shape -2 ; exponential yearly loss hazard rates of 0.05.

```
. stpower exponential 0.3 0.15, onesided power(0.9) aperiod(4) fperiod(2)
> ashape(-2) losshaz(0.05 0.05)
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions
Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Input parameters:

(omitted)

Accrual and follow-up information:

```
duration = 6.0000
follow-up = 2.0000
accrual = 4.0000 (exponential)
accrued(%) = 50.00 (by time t*)
t* = 3.6536 (91.34% of accrual)
lh1 = 0.0500
lh2 = 0.0500
```

Estimated sample sizes:

```
N = 194
N1 = 97
N2 = 97
```

Obtaining estimates of power or effect size

- Power determination:
specify sample size in `n()`
- Effect-size determination:
specify both sample size in `n()` and power in `power()` or,
specify both `n()` and prob. of a type II error in `beta()`;
not available with `stpower exponential`

Note that the value of the estimated effect size corresponding to the reduction in a hazard of the experimental group ($\Delta < 1$, $\ln(\Delta) < 0$, $\beta < 0$) is reported.

Example

Recall design B of a study comparing survivor functions using the log-rank test. Compute power for a fixed sample size of 148.

```
. stpower logrank 0.4, n(148)
Estimated power for two-sample comparison of survivor functions
Log-rank test, Freedman method
Ho: S1(t) = S2(t)
Input parameters:
    alpha =    0.0500  (two sided)
    s1 =    0.4000
    s2 =    0.6325
    hratio =    0.5000
    N =      148
    p1 =    0.5000
Estimated number of events and power:
    E =      72
    power =    0.8053
```


Example

Compute a minimal detectable value of the hazard ratio given 80% power and a sample size of 148.

```
. stpower logrank 0.4, n(148) power(0.8)
Estimated hazard ratio for two-sample comparison of survivor functions
Log-rank test, Freedman method
Ho: S1(t) = S2(t)
Input parameters:
      alpha =    0.0500   (two sided)
       s1 =    0.4000
       s2 =    0.6308
        N =      148
     power =    0.8000
        p1 =    0.5000
Estimated number of events and hazard ratio:
         E =       72
    hratio =    0.5028
```

- `stpower` allows obtaining results for multiple values of survival probabilities, hazard rates, and coefficients (must be enclosed in parentheses)
- `stpower` allows obtaining results for multiple values specified in `alpha()`, `power()` or `beta()`, `n()`, and `hratio()`
- `stpower logrank` and `stpower exponential` also allow obtaining results for multiple values specified in `p1()` or `nratio()`
- multiple values may be directly enumerated or specified as *numlist*
- results for multiple values of other options may be obtained by using `forvalues`; see example 5 in [ST] **stpower exponential**

- option `table` displays results in a table with default columns
- option `columns(colnames)` displays results in a table with *colnames* columns (see help files for the description of *colnames*)
- if multiple values per option are specified results are displayed in a table automatically
- use `table saving()` to save table data and use `list` to display results or `graph` to produce plots of results

Example

Display results from example of `stpower` exponential (design A) in a table

```
. stpower exponential 0.3 0.15, onesided power(0.9) table
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Power	N	N1	N2	H1	H2	H2-H1	Alpha*
.9	82	41	41	.3	.15	-.15	.05

* one sided

Example

Compute sample sizes for $\lambda_2 = 0.15$ and $\lambda_2 = 0.18$ from example of stpower exponential (design B)

```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.9) aperiod(4) fperiod(2)  
Note: input parameters are hazard rates.
```

Estimated sample sizes for two-sample comparison of survivor functions
Exponential test, hazard difference, conditional
Ho: h2-h1 = 0

Power	N	N1	N2	H1	H2	H2-H1	Alpha*
.9	136	68	68	.3	.15	-.15	.05
.9	232	116	116	.3	.18	-.12	.05

FP	AP+
2	4
2	4

* one sided

+ uniform accrual; 50.00% accrued by 50.00% of AP

Example

Control column width by using option colwidth()

```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.7(0.05)0.9) aperiod(4)  
> fperiod(2) colwidth(7)
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Power	N	N1	N2	H1	H2	H2-H1	Alpha*	FP	AP+
.7	74	37	37	.3	.15	-.15	.05	2	4
.75	86	43	43	.3	.15	-.15	.05	2	4
.8	98	49	49	.3	.15	-.15	.05	2	4
.85	114	57	57	.3	.15	-.15	.05	2	4
.9	136	68	68	.3	.15	-.15	.05	2	4
.7	126	63	63	.3	.18	-.12	.05	2	4
.75	146	73	73	.3	.18	-.12	.05	2	4
.8	166	83	83	.3	.18	-.12	.05	2	4
.85	194	97	97	.3	.18	-.12	.05	2	4
.9	232	116	116	.3	.18	-.12	.05	2	4

* one sided

+ uniform accrual; 50.00% accrued by 50.00% of AP

Example

Use option `columns()` to select specific columns to be displayed in a table.

```
. local columns power n ea eo la lo h1 h2 aperiod fperiod alpha
. stpower exponential 0.3 (0.15 0.18), onesided power(0.8 0.9) aperiod(4) fperiod(2)
> losshaz(0.05 0.05) colw(7 6 6 6 6 6 7) columns('columns')
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Power	N	E Ha	E Ho	L Ha	L Ho	H1	H2	AP+	FP	Alpha*
.8	108	56	58	13	12	.3	.15	4	2	.05
.9	148	76	78	18	18	.3	.15	4	2	.05
.8	182	99	100	22	20	.3	.18	4	2	.05
.9	252	137	140	29	30	.3	.18	4	2	.05

* one sided

+ uniform accrual; 50.00% accrued by 50.00% of AP

Example

Use option `parallel` to request results to be computed for pairs of values rather than for all possible combinations of values

```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.8 0.9) aperiod(4) fperiod(2)
> losshaz(0.05 0.05) colw(7 6 6 6 6 6 7) columns('columns') parallel
```

Note: input parameters are hazard rates.

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Power	N	E Ha	E Ho	L Ha	L Ho	H1	H2	AP+	FP	Alpha*
.8	108	56	58	13	12	.3	.15	4	2	.05
.9	252	137	140	29	30	.3	.18	4	2	.05

* one sided

+ uniform accrual; 50.00% accrued by 50.00% of AP

Note: options with multiple values must contain the same numbers of values if `parallel` is specified

Example

Display results sorted on values of power first and then on values of the experimental group hazard rate h2

```
. qui stpower exponential 0.3 (0.15 0.18), onesided power(0.7(0.05)0.9) aperiod(4)
> fperiod(2) colwidth(7) saving(mydata, replace)
. use mydata
. sort power h2
. list
```

	power	n	n1	n2	h1	h2	diff	alpha	fperiod	aperiod
1.	.7	74	37	37	.3	.15	-.15	.05	2	4
2.	.7	126	63	63	.3	.18	-.12	.05	2	4
3.	.75	86	43	43	.3	.15	-.15	.05	2	4
4.	.75	146	73	73	.3	.18	-.12	.05	2	4
5.	.8	98	49	49	.3	.15	-.15	.05	2	4
6.	.8	166	83	83	.3	.18	-.12	.05	2	4
7.	.85	114	57	57	.3	.15	-.15	.05	2	4
8.	.85	194	97	97	.3	.18	-.12	.05	2	4
9.	.9	136	68	68	.3	.15	-.15	.05	2	4
10.	.9	232	116	116	.3	.18	-.12	.05	2	4

- Recall the following example:

```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.8 0.9) aperiod(4)
> fperiod(2) losshaz(0.05 0.05) colw(7 6 6 6 6 6 7)
> columns(power n ea eo la lo h1 h2 aperiod fperiod alpha) parallel
```

- Recall the following example:

```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.8 0.9) aperiod(4)
> fperiod(2) losshaz(0.05 0.05) colw(7 6 6 6 6 6 7)
> columns(power n ea eo la lo h1 h2 aperiod fperiod alpha) parallel
```

- Go to **Statistics** > **Power and sample size** > **Exponential test**
- Main tab: fill in values for power and hazard rates, check “Treat *list as parallel” box and choose type of test to be one sided
- Accrual/Follow-up tab: fill in values for accrual and follow-up periods, and for loss to follow-up hazard rates
- Reporting tab: select columns to be displayed in the table and optionally fill in values for column widths
- Click **Submit** or **OK**

stpower exponential - Sample size and power for the exponential test

Main | **Accrual / Follow-up** | Reporting

Compute

Sample size 0.05 * Significance level

Power

Sample size: * Sample size

Power (specify as power or type II error)

0.8 0.9 * Power

0.2 * Type II error probability

Effect size and control group rate

Specify as:

Hazard rates

Survival probabilities

Reference time

Effect size and control group rate

0.5 * Hazard ratio

0.3 * Hazard rate in the control group

0.15 0.18 * Hazard rate in the experimental group

Sample allocations for control and experimental groups

Specify sample-size allocation as:

Proportion of subjects in control group

Ratio of experimental group to control group

0.5 * Proportion of subjects in the control group

Use the test of the difference between log hazards (log hazard-ratio test)

Use the unconditional approach

Treat *lists as parallel * Accepts list Help

One-sided test ▼ Type of test

? R []

OK Cancel Submit

stpower exponential - Sample size and power for the exponential test

Main Accrual / Follow-up Reporting

Length of the follow-up period, f

Accrual specification

Length of the accrual period, R

Exponential accrual

Specify accrual pattern under truncated exponential accrual

Proportion of subjects accrued by time t^* , $G(t^*)$

Proportion of the accrual period, t^*/R

Reference accrual time, t^*

Specify shape of the truncated exponential accrual distribution

Shape parameter

Loss to follow-up specification

Proportion of subjects lost to follow-up in the control and the experimental groups

Reference loss to follow-up time

Loss hazard rates in the control and the experimental groups

stpower exponential - Sample size and power for the exponential test

Main Accrual / Follow-up Reporting

Show detailed output

Display results in a table (default if "list" is specified on the "Main" tab)

Table specification

Table columns

Default columns

Select columns

Custom columns...

Suppress title

Suppress legend

7 6 6 6 6 7 Column widths

0 Separator every N lines


Save table data as dataset

Filename:

Browse...

Overwrite file if it already exists

OK Cancel Submit



```
. stpower exponential 0.3 (0.15 0.18), onesided power(0.8 0.9) aperiod(4) fperiod(2)
> loss haz(0.05 0.05) colw(7 6 6 6 6 6 7)
> columns(power n ea eo la lo h1 h2 aperiod fperiod alpha) parallel
Note: input parameters are hazard rates.
```

Estimated sample sizes for two-sample comparison of survivor functions

Exponential test, hazard difference, conditional

Ho: $h_2 - h_1 = 0$

Power	N	E Ha	E Ho	L Ha	L Ho	H1	H2	AP+	FP	Alpha*
.8	108	56	58	13	12	.3	.15	4	2	.05
.9	252	137	140	29	30	.3	.18	4	2	.05

* one sided

+ uniform accrual; 50.00% accrued by 50.00% of AP

Manually:

- specify *numlist* to compute powers for a range of values and use `saving()` to save results in a dataset
- use `graph twoway` to plot results
- use overlaid plots to produce multiple power curves

Automatic:

- use unofficial wrapper `stpowplot` to obtain simple, overlaid and separate graphs of power and other curves

Example

Produce a simple power curve as a function of sample size for a study from example of `stpower logrank` (design B)


```
. qui stpower logrank 0.4, n(50(10)200) saving(mypower, replace)

. use mypower

. line power n,
>   title("Power vs sample size")
>   subtitle("Log-rank test, Freedman method")
>   ytitle("Power") xtitle("Number of subjects")
>   xlabel(50(20)200, grid) ylabel(#10, grid)
>   note("Alpha=.05 (two sided); HR=.5; S1=.4; S2=.63; N2/N1=1")
```


Power vs sample size

Log-rank test, Freedman method



Alpha=.05 (two sided); HR=.5; S1=.4; S2=.63; N2/N1=1

```
. stpower logrank 0.4, n(50(10)200) yaxis(power) xaxis(n)
Estimating power ...
```



Other study parameters: Alpha = .05 (two sided); HR = .5; S1 = .4; S2 = .63; N2/N1 = 1



Example

Overlaid power curves as a function of sample size for the three values of hazard ratio


```
. qui stpower logrank 0.4, n(50(10)200) hratio(0.4 0.5 0.6)
> saving(mypower, replace)

. use mypower

. gr twoway line power n if hr==0.4 || line power n if hr==0.5
>                               || line power n if hr==0.6,
>   title("Power vs sample size")
>   subtitle("Log-rank test, Freedman method")
>   ytitle("Power") xtitle("Number of subjects")
>   xlabel(50(20)200, grid) ylabel(#10, grid)
>   note("Alpha = .05 (two sided); S1 = .4; N2/N1 = 1")
>   legend(label(1 "HR=0.4") label(2 "HR=0.5") label(3 "HR=0.6") rows(1))
```


Power vs sample size

Log-rank test, Freedman method



Alpha = .05 (two sided); S1 = .4; N2/N1 = 1

```
. stpowplot logrank 0.4, n(50(10)200) hratio(0.4 0.5 0.6) yaxis(power) xaxis(n)
> over(hr)
Estimating power ...
```



Other study parameters: Alpha = .05 (two sided); S1 = .4; N2/N1 = 1



Example


Produce power curves as a function of hazard ratio for four sample-size values

```
. qui stpower logrank 0.4, n(50(50)200) hratio(0.4(0.05)0.9) saving(mypower, replace)
. use mypower

. gr twoway line power hr if n==50 || line power hr if n==100
>     || line power hr if n==150 || line power hr if n==200,
>     title("Power vs hazard ratio")
>     subtitle("Log-rank test, Freedman method")
>     ytitle("Power") xtitle("Hazard ratio")
>     xlabel(0.4(0.05)0.9, grid) ylabel(#10, grid)
>     note("Alpha = .05 (two sided); S1 = .4; N2/N1 = 1")
>     legend(label(1 "N=50") label(2 "N=100") label(3 "N=150") label(4 "N=200"))
```




Power vs hazard ratio

Log-rank test, Freedman method



Alpha = .05 (two sided); S1 = .4; N2/N1 = 1


```
. stpowplot logrank 0.4, n(50(50)200) hratio(range 0.4 0.9 np 20) yaxis(power)
> xaxis(hr) over(n)
Estimating power ...
```



```
. stpowplot logrank 0.4, alpha(0.05 0.1) n(50(50)200) hratio(range 0.4 0.9)
> yaxis(power) xaxis(hr) over(n) by(alpha)
Estimating power ...
```

Power vs hazard ratio

Log-rank test, Freedman method




Other study parameters: S1 = .4; N2/N1 = 1



```
. stpowplot logrank 0.4, nratio(1 2) alpha(0.05 0.1) n(50(50)200) hratio(r 0.4 0.9)
> yaxis(power) xaxis(hr) over(n) by(alpha nratio) ylabel(#5, grid)
Estimating power ...
```

Power vs hazard ratio

Log-rank test, Freedman method



Other study parameters: S1 = .4



You can use `stpower` to

- estimate sample size, power, or minimal detectable effect size
- perform power analysis for two-sample log-rank tests, parametric (exponential survival) tests, and Cox PH models
- compute results for multiple values of study parameters and display them in a table
- build your own table of results
- save results in a dataset for further production of power and other curves

You can use `stpowlplot` to

- automatically generate plots of power and other curves
- produce overlaid plots using `over()`
- produce separate plots using `by()`

You can obtain `stpowlplot` by typing

```
. net from http://www.stata.com/users/ymarchenko/
. net describe stpowlplot
. net install stpowlplot
```

- Collett, D. 2003. *Modelling Survival Data in Medical Research*. London: Chapman & Hall/CRC.
- Freedman, L. S. 1982. Tables of the number of patients required in clinical trials using the logrank test. *Statistics in medicine* 1: 121–129.
- Hsieh, F. Y., and P. W. Lavori. 2000. Sample size calculations for the Cox proportional hazards regression models with nonbinary covariates. *Controlled Clinical Trials* 21: 552–560.
- Lachin, J. M. 1981. Introduction to sample size determination and power analysis for clinical trials. *Controlled Clinical Trials* 2: 93–113.
- Lachin, J. M. 2000. *Biostatistical Methods: The Assessment of Relative Risks*. New York: Wiley.

Lachin, J. M., and M. A. Foulkes. 1986. Evaluation of sample size and power for analysis of survival with allowance for nonuniform patient entry, losses to follow-up, noncompliance, and stratification. *Biometrics* 42: 507–519.

Rubinstein, L. V., M. H. Gail, and T. J. Santner. 1981. Planning the duration of a comparative clinical trial with loss to follow-up and a period of continued observation. *Journal of Chronic Diseases* 34: 469–479.

Schoenfeld, D. 1981. The asymptotic properties of nonparametric tests for comparing survival distributions. *Biometrika* 68: 316–319.

Schoenfeld, D. 1983. Sample-size formula for the proportional-hazards regression model. *Biometrics* 39: 499–503.